
TILASTOLLISET MENETELMÄT TÄHTITIETEESSÄ
IDL-harjoitus 2: Esimerkkivastaukset, Heikki Salo 27.3.2008

1. Luo N luvun otos välille [0, 1] tasaisesti jakaantuneita satunnaismuuttujia X (merkitään
Xi ∼ T AS (0, 1)) käyttäen IDL:n randomu-funktiota.

Kokeile esim. IDL> plot,randomu(seed,10000),psym=3

Tämä antaa plotin jossa y-akselina on satun-
naisluvun arvo (x-akselilla on luodun satunnais-
luvun jarjestysnumero).

- Plottaa histo_f-proseduurin avulla lukujen jakauma, jakamalla väli [0,1] esim. 10, 100,
1000,... väliin. Piirrä samaan kuvaan tasaisen jakauman teoreettinen tiheysfunktio. Kokeile eri
N:n arvoja.

Ohjelma idlharj2_tas01.pro lue eo. jakauman mukaisia satunnaislukuja ja plottaa luo-
tujen satunnaislukujen jakauman histogrammina. Ohessa esimerkkejä eri N:n ja jakovälien dx
arvoilla. Karkea arvio satunnaisuuden aiheuttamalle suhteelliselle vaihtelulle on ±1/

√
Nbin, jos-

sa Nbin ilmoittaa kuinka montaa satunnaislukua histogrammin yksi pylväs keskimäärin vastaa.
(arvio seuraa siitä että lukumäärä on Poisson-jakaantunut, keskiarvo Nbin, hajonta

√
Nbin; arvio

merkitty kuviin pystyviivalla).
N=1000, dx=0.01 -> 10/bin

N=10000, dx=0.01 -> 100/bin

N=100, dx=0.1 -> 10/bin

N=10000, dx=0.1 -> 1000/bin

Huom: histo_f normeeraa pylväiden yhteenlasketun pinta-alan yksikköpinta-alaan, eli se
approksimoi tiheysfunktiota (edellyttäen että annetut rajat kattavat koko jakauman (voidaan
ohittaa /noscale-keywordilla ⇒ plottaa lukumäärät kullakin jakovälillä ilman normeerausta).

;-----------------------------------------------------------
;idlharj2_tas01
;-----------------------------------------------------------
;yksinkertainen plottaus
program=’idlharj2_tas’
ps=-1
psdirect,program,ps
plot,randomu(seed,10000),psym=3
psdirect,program,ps,/stop

;-----------------------------------------------------------------
program=’idlharj2_tas01’
ps=-1
psdirect,program,ps

;tasainen jakauma valilla 0-1
;valitaan eri N = satunnaislukujen lukumaara ja plotataan histogrammi.
;histo_f laskee eri osavaleille osuvien arvojen lukumaarat
;ja normeeraa ne niin etta histogrammin pinta-ala = 1
;-> approksimoi tiheysfunktiota
;Jotta normeeraus oikein, jakauman on rajoituttava annettujen rajojen valiin
;histogram from x1 to x2 with step dx
;xx,yy return calculated values
;histo_f,x,x1,x2,dx,xx,yy,

nwin
!p.multi=[0,2,2]
!y.range=[0,2]
!y.style=1
!p.charsize=0.7

;-------------------------------------------------------------------
N=1000l
dx=0.01
x=randomu(seed,n)
title=’N=1000, dx=0.01 -> 10/bin’
histo_f,x,-.2,1.2,dx,xx,yy
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=title,xs=1

;plotataan samaan kuvaan arvio yksittaisten pylvaiden vaihteluista:
;
; DELTA(NBIN) sqrt(NBIN) 1
; -------------- = ------------- = ----------
; NBIN NBIN sqrt(NBIN)
;
xarg=[1,1]*1.1
yarg=xarg*0.+1
nbin=n*dx
dyarg=xarg*0.+sqrt(1./nbin)
oploterr,xarg,yarg,dyarg

;teoreeettinen tiheysjakauma
xteo=findgen(101)/100.
yteo=xteo*0.+1
oplot,xteo,yteo,lines=0

;-------------------------------------------------------------------
N=10000l
dx=0.01
x=randomu(seed,n)
title=’N=10000, dx=0.01 -> 100/bin’
histo_f,x,-.2, 1.2,dx,xx,yy
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=title,xs=1

nbin=n*dx
dyarg=xarg*0.+sqrt(1./nbin)

oploterr,xarg,yarg,dyarg
;-------------------------------------------------------------------
N=100l
dx=0.1
x=randomu(seed,n)
title=’N=100, dx=0.1 -> 10/bin’
histo_f,x,-.2, 1.2,dx,xx,yy
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=title,xs=1

nbin=n*dx
dyarg=xarg*0.+sqrt(1./nbin)
oploterr,xarg,yarg,dyarg

;-------------------------------------------------------------------
N=10000l
dx=0.1
x=randomu(seed,n)
title=’N=10000, dx=0.1 -> 1000/bin’
histo_f,x,-.2,1.2,dx,xx,yy
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=title,xs=1

nbin=n*dx
dyarg=xarg*0.+sqrt(1./nbin)
oploterr,xarg,yarg,dyarg

!p.multi=0
psdirect,program,ps,/stop

end

- Miten luot satunnaislukuja välilla [a, b], mika on luomasi otoksen keskiarvo ja keskihajonta?
Miten nämä suhtautuvat tasaisen jakauman teoreettiseen keskiarvoon (µ = b+a

2 ) ja keskihajon-
taan (σ = b−a

√
12

). Vaikuttaako N otoksen keskihajontaan?

Välille [a, b] tasaisesti jakautuneita lukuja saadaan muunnoksella

y = (b − a)x + a

jossa x on tasaisesti jakaantunut välille [0, 1].

IDL> a=10
IDL> b=15
IDL> print,mean((b-a)*randomu(seed,100)+a),(b+a)/2.

12.4469 12.5000
IDL> print,mean((b-a)*randomu(seed,100000)+a),(b+a)/2.

12.5032 12.5000
IDL> print,stdev((b-a)*randomu(seed,100)+a),(b-a)/sqrt(12.)

1.46380 1.44338
IDL> print,stdev((b-a)*randomu(seed,100000)+a),(b-a)/sqrt(12.)

1.44160 1.44338

Otoskeskiarvo ja otoshajonta eivät sinänsä riipu N:stä (niiden odotusarvo sama riippumatta
N:n arvosta), mutta satunnainen vaihtelu pienenee N:n kasvaessa.

- Totea käyttämällä eri N:n arvoja, että otoksen keskiarvo X̄ lähestyy tasaisen jakauman teoreet-
tista keskiarvoa µ otoksen N:n kasvaessa. Arvioi poikkeaman X̄−µ keskihajontaa (=”keskiarvon
keskivirhe”) luomalla M kappaletta N:n luvun otoksia, ja laskemalla saaduista X̄ j, j = 1, ...M
niiden otoshajonta σµ. Miten σµ suhtautuu teoreettiseen arvioon σ/

√
N, jossa σ on tasaisen

jakauman keskihajonta.
(Otoskeskiarvojen ja otoshajontojen laskemiseen IDL:ssä on mean ja stdev funktiot. Suo-

situs on käyttää uudempaa moment-funktiota, joka palauttaa keskiarvon, varianssin, vinokkuu-
den ja huipukkuuden)

otoksien lukumaara M=   10000

;-----------------------------------------------------------
;idlharj2_tas_otos.pro
;-----------------------------------------------------------
ps=0
program=’idlharj2_tas_otos’
psdirect,program,ps

;Tutkitaan tasaisesta jakaumasta otettujen satunnaislukujen keskiarvon
;suppenemista kohti teoreettista arvoa otoskoon kasvaessa.
;Tasaisen jakauman teoreettinen keskiarvo=mu, hajonta=sigma

;Valitaan eri otoskokoja N_tab=[N1, N2,...],
;lasketaan otoskeskiarvo xbar
;Erotuksen xbar-\mu hajontaa kutsutaan keskiarvon keskivirheeksi

;Teoreettinen arvio = sigma/sqrt(N)
;Tutkitaan ’kokeellisesti’: luodaan M kappaletta kutakin N luvun otosta
;ja taltioidaan xbar-mu kullekin M, ja lopuksi nain taltioitujen
;erotusten otoshajonta taulokkoon otossigma_tab

;valitaan mielivaltaiset jakauman ala- ja ylaraja
a=5. & b=10.
mu=0.5*(a+b) &sigma=(b-a)/sqrt(12.)

;otoskoot
N_tab=[100l,300l,1000l,3000l,10000l,30000l, 100000l]

;otoshajonnat tallennetaan taulukkoon
otossigma_tab=n_tab*0.

;kutakin otosta luodaan M kappaletta
M=20

for in=0,n_elements(n_tab)-1 do begin
N=n_tab(in)
xbar_tab=fltarr(m)
for im=0,m-1 do begin
x=randomu(seed,n)*(b-a)+a
xbar_tab(im)=mean(x)-mu

endfor

otossigma_tab(in)=stdev(xbar_tab)
endfor

nwin
plot,n_tab,otossigma_tab,/ylog,/xlog,$

title=’satunnaisluvut tasaisesta jakaumasta’,$
xtitle=’otoksen koko N’,ytitle=’otoskeskiarvon keskivirhe’,psym=6

oplot,n_tab,sigma/sqrt(n_tab),lines=2
stm=’(M=’+strtrim(string(m),2)+’)’
label_data,0.25,0.9,[’ keskivirhe:otos ’+stm,’ teoreettinen arvio’],$

psym=[6,0],lines=[0,2]
psdirect,program,ps,/stop

;-----------------------------------------------------
;havainnollistetaan viela keskiarvon keskivirheen merkitysta
;verrataan xbar-mu jakaumaa
;ja gaussista jakaumaa jonka hajonta sigma/sqrt(N)

program=’idlharj2_tas_otos2’
psdirect,program,ps

M=10000 ; monta otosta, jotta niista saataisiin hyva jakauma
N=10 ;otoskoolla ei valia (totea)

xbar=fltarr(m)
for im=0,m-1 do begin
x=randomu(seed,n)*(b-a)+a
xbar(im)=mean(x)-mu

endfor
otossigma=stdev(xbar_tab)

x1=-3*sigma/sqrt(n)
x2= 3*sigma/sqrt(n)
dx=(x2-x1)/100.
histo_f,xbar,x1,x2,dx,xx,yy

nwin
yr=[0,max(yy)*1.5]
plot,xx,yy,psym=10,xtitle=’xbar-mu’,ytitle=’pdf’,yr=yr,title=’otoksien lukumaara M=’+string(M)

;approksimoidaan gaussisella jakaumalla jonka
;keskiarvo mu_gauss=E(xbar-mu)=0
;hajonta sigma_gauss=sigma/sqrt(N)

sigma_gauss=sigma/sqrt(N)
oplot,xx,1./sqrt(2.*!pi*sigma_gauss^2)*exp(-0.5*(xx/sigma_gauss)^2),lines=2

stn=’(N=’+strtrim(string(n),2)+’)’
label_data,0.05,0.9,[’ otoksen keskivirheen jakauma ’+stn,’ teoreettinen (suuri N)’],$

psym=[0,0],lines=[0,2],size=.8

psdirect,program,ps,/stop
end

2.
Luo N kappaleen otos normaalijakaumaa noudattavia satunnaismuuttujia (Xi ∼ N(0, 1)) käyt-
täen IDL:n randomn-funktiota.

IDL> plot,randomn(seed,10000),psym=3

Tarkista otoksen keskiarvo ja hajonta. Kuinka luot normaalijakauman, jonka keskiarvo on µ ja
keskihajonta σ?

Lukuja saadaan muunnoksella
y = σx + µ

jossa x on normeerattua normaalijakaumaa noudattava muuttuja.

IDL> sigma=3
IDL> mu=1
IDL> print,mean(sigma*randomn(seed,10000)+mu)

0.946455
IDL> print,stdev(sigma*randomn(seed,10000)+mu)

2.98254

-Plottaa saamasi jakauma eri N:n arvoilla, yhdessä teoreettisen Gaussisen tiheysfunktion
kanssa.

p(x) =
1

√
2πσ2

exp
[
−

1
2σ2 (x − µ)2

]

N=       10000

-Plottaa otoksesi kumulatiivinen tiheysjakauma. (Vihje: sort-funktio antaa taulukon alkioi-
den suuruusjarjestyksen, eli x(sort(x) järjestää x:n alkiot suuruusjärjestykseen; Kokeile mita
saat kun kirjoitat “plot,x(sort(x)), lindgen(n)/(float(n)-1)”, jossa “n=n_elements(x)”
on otoksen alkioiden lukumäärä)

Vertaa Gaussisen jakauman teoreettiseen kumulatiiviseen tiheysfunktioon Φ(x)
(Muista gauss_pdf-funktio.)

;-----------------------------------------------------------
;idlharj2_gauss
;-----------------------------------------------------------
;Gaussisesti jakautuneet satunnaisluvut
;keskiarvo mu, hajonta sigma
;-----------------------------------------------------------

;yksinkertainen plottaus
;normeerattu normaalijakauma: mu=0, sigma=1
program=’idlharj2_gauss’
ps=0
psdirect,program,ps
plot,randomn(seed,10000),psym=3
psdirect,program,ps,/stop

;-----------------------------------------------------------------
;valitaan eri N = satunnaislukujen lukumaara ja plotataan
;histogrammi.+ verrataan teoreettiseen jakaumaan
program=’idlharj2_gauss1’
ps=0
psdirect,program,ps
nwin
!p.charsize=0.7

sigma=3.
mu=1.
N=10000l
title=’N=’+string(N)
x=sigma*randomn(seed,n)+mu

;histogrammi valilla -5*sigma -> 5*sigma
x1=mu-5*sigma
x2=mu+5*sigma
dx=(x2-x1)/100.
histo_f,x,x1,x2,dx,xx,yy
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=title,xs=1

;teoreeettinen tiheysjakauma

xteo=x1+(x2-x1)*findgen(1001)/1000.
yteo=1./sqrt(2.*!pi)/sigma*exp(-0.5*(xteo-mu)^2/sigma^2)
oplot,xteo,yteo,lines=0

psdirect,program,ps,/stop

;-----------------------------------------------------------------
;kumulatiivinen tiheysfunktio

sigma=3.
mu=1.
N=10000l
x=sigma*randomn(seed,n)+mu

program=’idlharj2_gauss2’
ps=0
psdirect,program,ps
nwin

; plotataan otoksen luvut suurrujarjestyksessa, vastaavaksi y:n
; arvoksi laitetaan 0/n,

plot,x(sort(x)), (lindgen(n)+1)/float(n),xtitle=’x’,$
ytitle =’kumulatiivinen tiheysfunktio: P(X_i < x)’,col=2

;teoreettinen, kayttaen IDL gauss_pdf-funktiota
xteo=findgen(101)/100.*10.-5.
yteo=gauss_pdf(xteo)

oplot,xteo*sigma+mu,yteo,col=3,psym=6

label_data,0.1,0.9,[’otokselle laskettu’,’teoreettinen’],col=[2,3],psym=[0,6],lines=[0,0]
psdirect,program,ps,/stop

end

3. Normaalijakaantuneita satunnaismuuttujia voidaan luoda tasaisen jakauman avulla seu-
raavasti: Valitaan

x1 ∼ T AS (0, 1) ja
x2 ∼ T AS (0, 1)

ja muodostetaan luvut

y1 =
√
−2 log x1 cos 2πX2 ja

y2 =
√
−2 log x1 sin 2πX2

Totea että menetelmä toimii (myös IDL:n randomn-funktio käytää tätä Box-Muller menetel-
mää).

Box-Muller menetelmˆ⁄llˆ⁄ luotuja satunnaislukuja

;-------------------------------------------------------------------------
;idlharj2_boxmuller
;-------------------------------------------------------------------------
;Gaussisesti jakautuneiden satunnaislukujen luonti Box-Muller menetelmalla
;-------------------------------------------------------------------------
program=’idlharj2_boxmuller’
ps=0
psdirect,program,ps,/color

;tasaisesti jakaantuneet
n=10000
x1=randomu(seed,n)
x2=randomu(seed,n)

;nama ovat normaalijakauman (mu=0, sigma=1)
;mukaisesti jakaantuneet
y1=sqrt(-2*alog(x1))*cos(2*!pi*x2)
y2=sqrt(-2*alog(x1))*sin(2*!pi*x2)

;y1 histogrammi valilla -5*sigma -> 5*sigma
;y2 vastaava (tarkista!)
histo_f,y2,-5.,5.,.05,xx,yy
nwin
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,xs=1,$

title=’Box-Muller menetelmÃďllÃď luotuja satunnaislukuja’

;teoreeettinen tiheysjakauma
xteo=-5.+10*findgen(1001)/1000.
yteo=1./sqrt(2.*!pi)*exp(-0.5*xteo^2)
oplot,xteo,yteo,lines=0,col=2,thick=3

psdirect,program,ps,/stop
end

4. Normaalijakaantuneita lukuja voidaan tuottaa myös keskeiseen raja-arvolauseeseen poh-
jautuen. Otetaan N >> 1 tasaisesti jakaantunutta lukua Xi ∼ T AS (0, 1) ja muodostetaan niiden
summa. Keskeisen raja-arvolauseen perusteella summan jakauma lähenee normaalijakaumaa
N(µ, σ), jossa µ = summan odotusarvo (nyt N

2 ) ja σ2 sen varianssin odotusarvo ( N
12 )

Erityisesti,
y =
(∑12

i=1 Xi

)
− 6 ∼ N(0, 1) (Miksi?)

Totea käytännössä millaisia normaalijakauman approksimaatioita saat vaihtelemalla sum-
mattavien lukujen määriä.

Nsum=       2

Huom: eo arvolla N = 12 saatujen lukujen arvoalue on −6 ≤ X ≤ 6. Arvioi kuinka suu-
ri puute tämä on (eli miten suurella todennäköisyydellä oikea normaalijakauma N(0,1) antai-
si lukuja taman valin ulkopuolella) Karkean arvion voit tehda luomalla suuren määrän sa-
tunnaislukuja randomn-funktion avulla ja katsomalla moniko on tämän välin ulkopuolella.

IDL> n=1000000l
IDL> x=randomn(seed,n)
IDL> ind=where(abs(x) ge 1, count) & print,count/float(n)

0.317306
IDL> ind=where(abs(x) ge 2, count) & print,count/float(n)

0.0456940
IDL> ind=where(abs(x) ge 3, count) & print,count/float(n)
0.00268100

IDL> ind=where(abs(x) ge 4, count) & print,count/float(n)
6.10000e-05

IDL> ind=where(abs(x) ge 5, count) & print,count/float(n)
0.00000

IDL> ind=where(abs(x) ge 6, count) & print,count/float(n)
0.00000

q;-------------------------------------------------------------------------
;idlharj2_kral_tas
;-------------------------------------------------------------------------
;summataan Nsum tasaisesti jaantunutta lukuja
;keskeinen rajarvolause -> summa normaalisesti jakaantunut kun Nsum suuri
;-------------------------------------------------------------------------
program=’idlharj2_kral_tas’ & ps=0
psdirect,program,ps,/color

Nsumtab=[2,4,8,16]
!p.multi=[0,2,2]
Notos=100000l

for i=0,n_elements(nsumtab)-1 do begin
nsum=nsumtab(i)
xsum=fltarr(notos)
for j=0,nsum-1 do begin
xsum=xsum+randomu(seed,notos)

endfor

;teoreeettinen tiheysjakauma
;kaytetaan sopivien plottirajojen arviointiin

mu=nsum/2.
sigma=sqrt(nsum/12.)
xala=mu-5*sigma & xyla=mu+5*sigma &dx=sigma*.05

histo_f,xsum,xala,xyla,dx,xx,yy
nwin
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=’Nsum=’+string(Nsum),xs=1
xteo=(findgen(1001)/1000.*10-5.)*sigma+mu
yteo=1./sqrt(2.*!pi*sigma^2)*exp(-0.5*(xteo-mu)^2/sigma^2)
oplot,xteo,yteo,lines=0,col=2,thick=3

endfor
psdirect,program,ps,/stop
!p.multi=0

end

5. Luodaan exponentiaalisesti jakaantuneita satunnaislukuja tasaisesti jakaantuneiden lu-
kujen R ∼ T AS (0, 1) avulla.

f (x) = 1
c e−x/c kun x > 0 (x̄ = c)

Asetetean Ri =
∫ Xi

0
f (x)dx ⇒ Xi = −c log Ri

Totea että menetelmä toimii, vertaamalla sitä IDL:n randomu-funktion avulla luotuhin eks-
ponentiaalisestijakaantuneisiin lukuihin (kts ?randomu), sekä teoreettiseen tiheysfunktioon.

exponentiaalisesti jakaantuneet satunnaisluvut

;-------------------------------------------------------------------------
;idlharj2_expo
;-------------------------------------------------------------------------
;luodaan eksponentiaalisesti jaantumeita satunnaislukuja
;tasaisen jakauman avulla
;-------------------------------------------------------------------------
program=’idlharj2_expo’
ps=0
psdirect,program,ps,/color
N=100000
c=2 ; exponentiaalijakauman keskiarvo

;luodaan jakauma
r=randomu(seed,N)
x=-c*alog(r)

;histogrammi
histo_f,x,0.,6*c,0.1*c,xx,yy
nwin
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=’exponentiaalisesti jakaantuneet satunnaisluvut’,col=2

xyouts,0.25,0.9,/normal,’verrattu teoreettiseen jakaumaan’
xteo=findgen(1001)/1000.*10*c
yteo=1./c*exp(-xteo/c)
oplot,xteo,yteo,lines=0,col=3,thick=3
psdirect,program,ps,/stop

;-------------------------------------------------
;verrataan idl-randomu_function avulla saatuihin
program=’idlharj2_expo1’
psdirect,program,ps,/color
N=100000
c=2
r=randomu(seed,N)
x=-c*alog(r)
histo_f,x,0.,6*c,0.1*c,xx,yy
nwin
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,col=2,title=’exponentiaalisesti jakaantuneet satunnaisluvut’

;idl: funktiolla
y=randomu(seed,N,gamma=1)*c
histo_f,y,0.,10*c,0.1*c,xx,yy
oplot,xx,yy,col=3
xyouts,0.25,0.9,/normal,’verrattu IDL:n funktiolla generoituihin’
psdirect,program,ps,/stop

end

6. Demonstroi keskeisen raja-arvolauseen pätevyys summaamalla eksponentiaalisesti ja-
kaantuneita satunnaislukuja. Luennoilla annetussa esimerkissä (Wall-Jenkins Kuva 2.6) ver-
rattiin 2, 4, ja 16 eksponenttijakaumasta otetun luvun keskiarvon jakaumia Gaussiseen jakau-
mafunktioon. Käytä näitä samoja arvoja.

;-------------------------------------------------------------------------
;idlharj2_kral_tas
;-------------------------------------------------------------------------
;summataan Nsum eksponentiaalisesti jaantunutta lukuja
;keskeinen rajarvolause -> summa normaalisesti jakaantunut kun Nsum suuri
;-------------------------------------------------------------------------
program=’idlharj2_kral_expo’
ps=0
psdirect,program,ps,/color
!p.multi=[0,2,2]
nwin

Nsumtab=[2,4,8,16]
Notos=100000l
c=2.

for i=0,n_elements(nsumtab)-1 do begin
nsum=nsumtab(i)
xsum=fltarr(notos)
for j=0,nsum-1 do begin

;idl:n funktion avulla
x1=c*randomu(seed,notos,gamma=1)
xsum=xsum+x1

endfor

;teoreeettinen tiheysjakauma
;kaytetaan sopivien plottirajojen arviointiin
;eksponenttijakauman mu=c, sigma=c

mu=nsum*c
sigma=sqrt(nsum)*c
xala=mu-5*sigma
xyla=mu+5*sigma
dx=sigma*.05

histo_f,xsum,xala,xyla,dx,xx,yy
plot,xx,yy,psym=10,xtitle=’x’,ytitle=’pdf’,title=’Nsum=’+string(Nsum),xs=1

xteo=(findgen(1001)/1000.*10-5.)*sigma+mu
yteo=1./sqrt(2.*!pi*sigma^2)*exp(-0.5*(xteo-mu)^2/sigma^2)
oplot,xteo,yteo,lines=0,col=2,thick=3

endfor
psdirect,program,ps,/stop
!p.multi=0

end

7. Poisson jakauman luonti.

Poisson jakauma kuvaa prosessia, jossa aikayksikköä kohti tapahtuu keskimäärin λ tapah-
tumaa (raja-arvo binomijakaumasta, kun toistokokeen yksittäisessä yrityksessä on hyvin pieni
onnistumistodennakoisyys p, mutta yritysten määrä N on suuri, λ = pN). Tällöin tapahtuneiden
tapausten lukumäärä X noudattaa jakaumaa.

p(x, λ) =
λx

x!
e−λ x = 0, 1, 2, ...

Poisson prosessissa tapahtumien välinen odotusaika on eksponentiaalisesti jakaantunut. Kos-
ka tapahtumia on keskimäärin λ/aikayksikkö, niin odotusajan w keskiarvo on w̄ = 1/λ aikayk-
sikköä. Siten w:n jakauma on

f (w) =
1
w̄

e−w/w̄ = λ e−λw

Posson jakaantuneita satunnaislukuja voidaan tuottaa summaamalla eo. eksponentiaalija-
kauman mukaisesti jakaantuneita odotusaikoja Wi, kunnes niiden summa ylittaa aikayksikön.
Merkitään summattujen odotusaikojen lukumäärää X + 1 : ll Tällöin tiedetään, että aikayksikön
kuluessa tapahtui X tapausta. Näin luotujen X:n arvojen jakauma noudattaa Poisson-jakaumaa.

Tee ohjelma joka tuottaa Poisson jakautuneita satunnaislukuja käyttäen a) IDL randomu-
ohjelmaa ja b) edellä kuvattua menetelmää. Vertaa teoreettiseen jakaumaan. Valitse esim. λ = 5.

Poisson with lambda= 5

0 5 10 15 20
x

0.00

0.05

0.10

0.15

0.20

p(
x)

theoretical

idl randomu

directly

18



Poisson process with lambda= 5

;-----------------------------------------------------------
;idlharj2_poisson_rnd
;HS 050308
;-----------------------------------------------------------
program=’idlharj2_poisson_rnd’
ps=0
psdirect,program,ps,/color

;poisson jakauma:
lambda=5
x=findgen(21)
px=exp(-lambda)*lambda^x/factorial(x)

nwin
plot,x,px,xtitle=’x’,ytitle=’p(x)’,$

title=’Poisson with lambda=’+string(lambda,’(i2)’),yr=[0,max(px)*1.2]

;random number using IDL
r=randomu(seed,100000,poisson=lambda)
histo_f,r,-0.5,21,1,xx,yy
oplot,xx,yy,psym=10,col=2

;directly
;Poisson process with lambda events/timeunit
;waiting time between successive events is exponentially distributed,
;with a mean waiting time 1/lambda
;add up X+1 exponentially distributed waiting times, until the sum
;exceed timeunit -> X is the number of events during timeunit
; X is Poisson distributed.

rr=fltarr(100000)
for i=0l,n_elements(rr)-1 do begin
z=-1./lambda*alog(randomu(seed,20))
zc=total(z,/cum) ;cumulative sum
ind=where(zc gt 1)
rr(i)= ind(0) ;number of event during the time interval

;note first zc > 1 -> ind(0)=0 (x+1=1, x=0)

; second zc > 2 -> ind(0)=1 (x+1=2, x=1)
; etc zx > x+1 -> ind(0)=x

endfor
histo_f,rr,-0.5,21,1,xx,yy
oplot,xx,yy,psym=10,col=3

col1=1
if(!d.name eq ’PS’) then col1=0
label_data,0.5,0.9,[’theoretical’,’idl randomu’,’directly’],lines=[0,0,0],col=[col1,2,3]
psdirect,program,ps,/stop

;-------------------------------------------------------------
;illustration
;show 11 trials, each event until TIME> UNIT TIME
;number of event poisson distributed
program=’idlharj2_poisson_rnd1’
psdirect,program,ps,/color

nwin
plot,lindgen(2),/nod,xr=[-.1,3],xs=1,yr=[-1,11],ys=1,xtitle=’TIME’,$

ytitle=’TRIAL #’,$
title=’Poisson process with lambda=’+string(lambda,’(i2)’)

for i=0l,10 do begin
z=-1./lambda*alog(randomu(seed,100))
zc=total(z,/cum) ;cumulative sum
ind=where(zc lt 1,count)
if(count gt 0) then ind=lindgen(count)
if(count gt 0) then oplot,zc(ind),zc(ind)*0+i,psym=6,col=2,syms=1
oplot,zc(count)*[1,1],[i,i],psym=1,col=3

add=’’
if(i eq 10) then add=’ EVENTS/UNIT TIME’

xyouts,zc(count)+.1,i,string(count,’(i2)’)+add
oplot,[0,zc(count)],[i,i],lines=1,col=3
if(count gt 0) then oplot,[0,zc(count-1)],[i,i],lines=1

endfor
oplot,[1,1],[-1,11],lines=2

;insert of waiting time distribution
x=findgen(500)/100.*1./lambda
fx=lambda*exp(-x*lambda)
plot,x,fx,pos=[0.65,0.5,0.9,0.8],/noerase,xtitle=’waiting time’,ytitle=’p’
oplot,[1.,1.]/lambda,[0,10],lines=2,col=2

psdirect,program,ps,/color,/stop
end

